

CHAPTER 5 Generating Random Numbers

The generation of random numbers is done via LET subcommands. For example, the statement

```
LET X = UNIFORM RANDOM NUMBERS FOR I = 1 1 25
```

generates a variable X that contains uniform random numbers in the first 25 rows. The output from the random number generation is always a variable, never a parameter or function. The term (or terms) before RANDOM NUMBERS identifies the distribution for which random numbers are being generated. The first number after the "I =" clause denotes the row position of the first random number, the second number denotes the row increment, and the third number is the last row position for which a random number is generated. The start position and increment are usually 1. Any skipped rows are set to zero for a new variable and left as is for a previously created variable. Also, "I" must be used for the dummy index. Unpredictable results can occur if another name is used. As another example, the statement

```
LET Y = NORMAL RANDOM NUMBERS FOR I = 3 2 21
```

generates a normal random number for every other row, starting with row 3, until the 21st row of X has been completed. The following list was generated with this command.

```
0.0000000E+00  
0.0000000E+00  
-0.2430273E+00  
0.0000000E+00  
-0.8409334E+00  
0.0000000E+00  
-0.1038428E+00  
0.0000000E+00  
0.4185625E+00  
0.0000000E+00  
0.2635231E+00  
0.0000000E+00  
0.8981169E+00  
0.0000000E+00  
0.3407212E-01  
0.0000000E+00  
0.1587882E+01  
0.0000000E+00  
0.3888193E+00  
0.0000000E+00  
-0.4696768E+00
```

Some distributions represent a family of distributions. In this case, one or more parameters need to be specified (via the LET command) before generating the random numbers. For example

```
LET GAMMA = 2.5  
LET Z = PARETO RANDOM NUMBERS FOR I = 1 1 100
```

In addition, the Weibull, extreme value type I (or Gumbel), extreme value type II (or Frechet), and generalized Pareto distributions can be based on either the minimum or maximum order statistic. The

command SET MINMAX <1/2> is required before generating random numbers for these distributions. Failure to specify the MINMAX parameter results in a large number of error messages being printed since there is no default value for MINMAX. A value of 1 specifies the minimum order statistic while a value of 2 specifies the maximum order statistic. Currently, the generalized Pareto distribution is only supported for the maximum order statistic (i.e., enter SET MINMAX 2).

```
LET GAMMA = 2.5
SET MINMAX 1
LET Z = PARETO RANDOM NUMBERS FOR I = 1 1 100
```

DATAPLOT generates random numbers for the standard form of a distribution. Many of these distributions have location and scale parameters that are set to 0 and 1 respectively. However, it is straightforward to generate random numbers for the nonstandard forms. The following example shows how to generate random numbers for a normal distribution with a mean of 5 and a standard deviation of 10.

```
LET M = 5
LET SD = 10
LET Y = NORMAL RANDOM NUMBERS FOR I = 1 1 100
LET Y = M + SD*Y
```

For all of the distributions for which DATAPLOT can generate random numbers, it can also generate the cumulative distribution function, the probability density (or mass) function, and the percent point function. These are documented in chapter 8 (Probability Library Functions) of this volume. The documentation for the probability density function specifies the distributional form of the function and what the scale and location parameters, if any, are. Probability plots can also be generated for these distributions. This is documented under the PROBABILITY PLOT command in chapter 2 of Volume I.

The algorithm used to generate uniform random numbers (the other distributions can be generated based on the uniform distribution) is a Fibonacci generator as defined by George Marsaglia in “Comments on the Perfect Random Number Generator” (unpublished notes from Washington State University).

An important feature of the Marsaglia uniform random number generator is that the same random numbers are generated regardless of the type of computer being used. This is possible because an internal variable, specified during the installation of DATAPLOT, controls the sequence of numbers generated. The internal variable is typically set to 32. However, hosts with a word size less than 32 bits will need to decrease this number since the internal variable cannot be greater than the number of bits per word.

The seed for the random number generator can be set with the SEED command (documented in the Support chapter of Volume I).

The following is a list of the available random number generators.

Distributions requiring no parameters

NORMAL RANDOM NUMBERS	Generates standard normal (N(0,1)) random numbers.
UNIFORM RANDOM NUMBERS	Generates uniform (0,1) random numbers.
LOGISTIC RANDOM NUMBERS	Generates logistic random numbers.
DOUBLE EXPONENTIAL RANDOM NUMBERS	Generates double exponential random numbers.
CAUCHY RANDOM NUMBERS	Generates Cauchy random numbers.
SEMI-CIRCULAR RANDOM NUMBERS	Generates semi-circular random numbers.
TRIANGULAR RANDOM NUMBERS	Generates triangular random numbers.
LOGNORMAL RANDOM NUMBERS	Generates lognormal random numbers.
HALFNORMAL RANDOM NUMBERS	Generates halfnormal random numbers.
EXPONENTIAL RANDOM NUMBERS	Generates exponential random numbers.
EXTREME VALUE TYPE 1 RANDOM NUMBERS	Generates extreme value type 1 random numbers.

FRECHET RANDOM NUMBERS	Generates extreme value type 2 random numbers.
PARETO RANDOM NUMBERS	Generates Pareto random numbers.
INVERSE GAUSSIAN RANDOM NUMBERS	Generates inverse gaussian random numbers.
RECIPROCAL INVER GAUSSIAN RAND NUMB	Generates reciprocal inverse gaussian random numbers.
FATIGUE LIFE RANDOM NUMBERS	Generates fatigue life random numbers.
WALD RANDOM NUMBERS	Generates Wald random numbers.
GENERALIZED PARETO RANDOM NUMBERS	Generates Generalized Pareto random numbers.

Distributions requiring the parameters P and N

Enter the commands LET P = <value> and LET N = <value> before generating the random numbers. The parameter P is the proportion of successes and the parameter N is the number of items in the sample. The value of P should be a real number between 0 and 1. The value of N should be an integer between 1 and the maximum number of rows for a variable. This defaults to 10,000 on most installations. However, entering a DIMENSION command changes the maximum number of rows for a variable. A non-integer value of N will be truncated to an integer value.

BINOMIAL RANDOM NUMBERS	Generates binomial random numbers.
-------------------------	------------------------------------

Distributions requiring the parameters P and K

Enter the commands LET P = <value> and LET K= <value> before generating the random numbers. The parameter P is the proportion of successes and the parameter K is the number of successes. The value of P should be a real number between 0 and 1. The value of K should be a positive real number.

NEGATIVE BINOMIAL RANDOM NUMBERS	Generates negative binomial random numbers.
----------------------------------	---------------------------------------------

Distributions requiring the parameter P

Enter the command LET P = <value> before generating the random numbers. The parameter P is the proportion of successes. The value of P should be a real number between 0 and 1.

GEOMETRIC RANDOM NUMBERS	Generates geometric random numbers.
--------------------------	-------------------------------------

Distributions requiring the parameter C

Enter the command LET C = <value> before generating the random numbers. The parameter C is a shape parameter for the distribution. The value of C can be any positive real number for the power function distribution.

POWER FUNCTION RANDOM NUMBERS	Generates power function random numbers.
-------------------------------	------------------------------------------

Distributions requiring the parameters K, N, and M

Enter the commands LET K= <value>, LET M = <value>, and LET N = <value> before generating the random numbers. The parameter K is the number of items sampled, M is the total number of items, and N is the number of marked items. All 3 parameters must be non-negative integers and N and K must both be less than M. Non-integer values of these parameters will be truncated to integer value.

HYPERGEOMETRIC RANDOM NUMBERS	Generates hypergeometric numbers.
-------------------------------	-----------------------------------

References

The following are excellent sources for information on a large range of probability distributions.

“Univariate Discrete Distributions,” Second Edition, Johnson, Kotz, and Kemp, John Wiley and Sons, 1992.


“Continuous Univariate Distributions - 1,” Second Edition, Johnson, Kotz, and Balakrishnan, John Wiley and Sons, 1992.

“Continuous Univariate Distributions - 2,” Second Edition, Johnson, Kotz, and Balakrishnan, John Wiley and Sons, 1994.

“Statistical Distributions,” Second Edition, Evans, Hastings, and Peacock, John Wiley and Sons, 1993.

PROGRAM 1

```
LET Y = NORMAL RANDOM NUMBERS FOR I = 1 1 1000
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
TITLE 1000 NORMAL RANDOM NUMBERS
HISTOGRAM Y
TITLE NORMAL PROBABILITY PLOT
NORMAL PROBABILITY PLOT Y
TITLE NORMAL RANDOM NUMBERS (M=50, S=20)
LET YNEW = 50 + 20*Y
HISTOGRAM YNEW
LET Z = CAUCHY RANDOM NUMBERS FOR I = 1 1 1000
TITLE NORMAL PROBABILITY PLOT (CAUCHY NUMBERS)
NORMAL PROBABILITY PLOT Z
END OF MULTIPLY
```



PROGRAM 2

```

MULTILOT 2 2; MULTILOT CORNER COORDINATES 0 0 100 100
LET Y = UNIFORM RANDOM NUMBERS FOR I = 1 1 1000
TITLE UNIFORM RANDOM NUMBERS
HISTOGRAM Y
LET Y = EXPONENTIAL RANDOM NUMBERS FOR I = 1 1 1000
TITLE EXPONENTIAL RANDOM NUMBERS
HISTOGRAM Y
LET GAMMA = 2
SET MINMAX 2
TITLE WEIBULL RANDOM NUMBERS (GAMMA=2)
LET Y = WEIBULL RANDOM NUMBERS FOR I = 1 1 1000
HISTOGRAM Y
LET Y = FRECHET RANDOM NUMBERS FOR I = 1 1 1000
TITLE FRECHET RANDOM NUMBERS (GAMMA=2)
HISTOGRAM Y
END OF MULTILOT
    
```

